University of Nebraska at Omaha

From the SelectedWorks of Yuliya Lierler

July, 2017

Answer Set Programming Paradigm

Yuliya Lierler

Available at: https://works.bepress.com/yuliya_lierler/72/
Handout on Answer Set Programming Paradigm

Yuliya Lierler
University of Nebraska Omaha

Introduction

Answer set programming paradigm (ASP) is a form of declarative programming oriented towards difficult combinatorial search problems. It belongs to the group of so called constraint programming languages. The idea of ASP is to represent a given computational problem by a program whose answer sets correspond to solutions, and then use an answer set solver to generate answer sets for this program. In this note we discuss the methodology of answer set programming as well as the use of software systems for computing answer sets. First, a graph coloring problem is utilized to illustrate the use of answer set programming in practice. Then, solutions to Hamiltonian cycle and to \( n \)-queens problems are presented. Across the handout you are given problems to solve. This handout is self-contained: you are given all the definitions and links that are required in constructing solutions.

In the text italics is primarily used to identify concepts that are being defined. Some definitions are identified by the word Definition.

In this course we will use the answer set system CLINGO\(^1\) that incorporates answer set solver CLASP\(^1\) with its front-end grounder GRINGO\(^1\) (user guide is available online at https://sourceforge.net/projects/potassco/files/guide/2.0/guide-2.0.pdf/download). You may access system CLINGO via web interface available at https://potassco.org/clingo/run/ or download an executable for CLINGO version 5 from the url listed at footnote 1.

Answer set programming practitioners develop applications that rely on ASP languages, which allow variables. Yet, common ASP solvers, including CLASP (a subsystem of CLINGO) process propositional logic programs only. We now re-introduce such programs and restate the definition of an answer set in the form convenient for this part of the course.

\(^1\)https://potassco.org/clingo/.
1 Formal preliminaries: Propositional logic programs

A propositional logic program is a finite set of rules of the form

\[ a_0 \leftarrow a_1, \ldots, a_m, \neg a_{m+1}, \ldots, \neg a_n, \neg \neg a_{n+1}, \ldots, \neg \neg a_p, \]  

(1)

where \( a_0 \) is a propositional atom or symbol \( \perp \); \( a_1, \ldots, a_p \) are propositional atoms. Note that rule (1) contains "doubly negated" atoms — a construct that you have not seen before. This construct is used to formalize so called choice rules that are frequently used by the ASP practitioners. Here we present the definition of an answer set for programs composed of rules (1) and then discuss in which sense they capture choice rules.

A rule (1) is normal when \( n = p \). A program is normal when it is composed of normal rules. The left hand side expression of rule (1) is called the head. The right hand side is called the body. Expression

\[ a_1, \ldots, a_m \]

constitutes positive part of the body. We call rule (1)

- a fact when its body is empty (we then drop \( \leftarrow \) from a rule);
- a constraint when its head is symbol \( \perp \) (we then drop \( \perp \) from a rule).

We call a program definite when it is composed of rules of the form

\[ a_0 \leftarrow a_1, \ldots, a_m. \]  

(2)

For instance,

\[ p \]

\[ r \leftarrow p, q \]  

(3)

and

\[ p \leftarrow \neg q \]

\[ q \leftarrow \neg r \]  

(4)

are both normal programs, where program (3) is also definite.

A set \( X \) of atoms satisfies a definite rule

\[ a_0 \leftarrow a_1, \ldots, a_m \]

(5)

when \( a_0 \in X \) whenever \( \{a_1, \ldots, a_m\} \subseteq X \). We say that a set \( X \) of atoms satisfies definite program \( \Pi \) if it satisfies every rule of \( \Pi \). For example, sets \( \{p\} \), \( \{p, r\} \), and \( \{p, q, r\} \) satisfy definite program (3).
Definition 1. The reduct $\Pi^X$ of a logic program $\Pi$ relative to a set $X$ of atoms is the set of rules (5) for all rules (1) in $\Pi$ such that $a_{m+1}, \ldots, a_n \not\in X$ and $\{a_{n+1}, \ldots, a_p\} \subseteq X$. A set $X$ of atoms is an answer set of the program $\Pi$ if $X$ is minimal among the sets of atoms that satisfy $\Pi^X$.

For instance, let $\Pi_1$ denote program (4) and $X_1$ denote set $\{q\}$. The reduct $\Pi_1^{X_1}$ consists of a single fact $q$. Set $\{q\}$ is the minimal set satisfying $\Pi_1^{X_1}$. Consequently, $X_1$ is an answer set of $\Pi_1$. Let $X_2$ denote set $\{p\}$. The reduct $\Pi_1^{X_2}$ consists of two facts $p$ and $q$. Set $\{p, q\}$ is the minimal set satisfying $\Pi_1^{X_2}$. Thus, $\{p\}$ is not an answer set of $\Pi_1$.

Answer sets semantics by default follows closed world assumption (CWA) – presumption that what is not currently known to be true is false. Let a program consist of the rule

$$p \leftarrow \text{not not } p.$$  

This program has two answer sets $\emptyset$ and $\{p\}$. This rule can be used to “eliminate” CWA for an atom $p$. Intuitively, it states that an atom $p$ may be a part of an answer set. This extension of logic programs is essential. Choice rules of the form

$$\{a_0\} \leftarrow a_1, \ldots, a_m, \text{not } a_{m+1}, \ldots, \text{not } a_n,$$  

are important constructs of common ASP dialects. A choice rule (7) can be seen as an abbreviation for a rule with doubly negated atoms of the form

$$a_0 \leftarrow a_1, \ldots, a_m, \text{not } a_{m+1}, \ldots, \text{not } a_n, \text{not not } a_0.$$  

Here we adapt this abbreviation. The simplest choice rule

$$\{p\}$$

corresponds to rule (6).

Consider a constraint $\leftarrow p$. Extending program (3) by this rule will result in a program that has no answer sets. In other words, constraint $\leftarrow p$ eliminates the only answer of (3). It is convenient to view any constraint

$$\leftarrow a_1, \ldots, a_m, \text{not } a_{m+1}, \ldots, \text{not } a_n, \text{not not } a_{p+1}, \ldots, \text{not } a_p,$$  

as a clause (a disjunction of literals)

$$-a_1 \lor \cdots \lor -a_m \lor a_{m+1} \lor \cdots \lor a_n \lor -a_{n+1} \lor \cdots \lor -a_p.$$  

3
Then, we can state the general property about constraints: answer sets of a program satisfy the propositional logic formula composed of its constraints (here (i) the notion of satisfaction is as understood classically in propositional logic and and (ii) an answer set is associated with an interpretation in an intuitive manner). Furthermore, for a program $\Pi$ and a set $\Gamma$ of constraints the answer sets of $\Pi \cup \Gamma$ coincide with the answer sets of $\Pi$ that satisfy $\Gamma$. Consequently, constraints can be seen as elements of classical logic in logic programs.

**Classical negation in programs** The textbook [1] immediately introduces programs that are of more complex form even in propositional case. Indeed, it allows additional connective $\neg$ (classical negation). So that basic entities of a program are literals (a *literal* is either an atom $a$ or an atom proceeded with classical negation $\neg a$) In turn, the concept of an answer set is defined over the consistent sets of literals whereas here we defined an answer set over the sets of atoms. Yet, it is easy to *simulate* classical negation in the simpler form of programs that we consider here.

Classical negation can always be eliminated from a program by means of auxiliary atoms and additional constraints. Indeed, given a program composed of rules of the form

$$ l_0 \leftarrow l_1, \ldots, l_m, \neg l_{m+1}, \ldots, \neg l_n $$

(10)

so that $l_0$ is a literal or $\bot$ and $l_1 \ldots l_n$ are literals, we can replace an occurrence of any literal of the form $\neg a$ with a fresh atom $a'$ and add a constraint to this program in the form

$$ \leftarrow a, \neg a'. $$

The answer sets of this new program as defined in this handout are in one to one correspondence with the answer sets as defined in the textbook. In particular, by replacing atoms of the form $a'$ by $\neg a$ we obtain the textbook answer sets.

2 Formalization of Graph coloring problem by means of a Propositional logic program

Consider a *graph coloring* problem $GC$:

A 3-coloring of a directed graph $(V, E)$ is a labeling of its vertexes with at most 3 colors (named, 1, 2 and 3) such that no two vertexes sharing an edge have the same color.
For instance, see two specific graphs $G_1$ and $G_2$ in Figure 1. It is easy to see that there are six distinct 3-colorings for graph $G_1$ including the following:

assigning color 1 to vertexes $a$ and $c$, color 2 to vertex $b$, and color 3 to vertex $d$ forms a 3-coloring of $G_1$.

We denote this 3-coloring of graph $G_1$ by $S_{G_1}$. Graph $G_2$ has no 3-colorings.

A solution to this problem can be described by a set of atoms of the form $c_{vi} (v \in V$ in a given graph $(V,E)$ and $i \in \{1, 2, 3\}$); including $c_{vi}$ in the set indicates that an assertion that vertex $v$ is assigned color $i$. A solution is a set $X$ satisfying the following conditions:

1. Each vertex must be assigned a color
2. A vertex may not be assigned more that one color
3. Vertexes sharing an edge must be assigned a distinct color.

A following propositional logic program under answer set semantics encodes these specifications so that its answer sets are in one to one correspondence with the 3-colorings of a given graph $(V,E)$:

\begin{align*}
\{c_{vi}\} & \quad (v \in V, 1 \leq i \leq 3) \\
\leftarrow not \ c_{v1}, not \ c_{v2}, not \ c_{v3}. & \quad (v \in V). \\
\leftarrow c_{vi}, c_{vj} & \quad (v \in V, 1 \leq i < j \leq 3), \\
\leftarrow c_{vi}, c_{wi} & \quad (\{v, w\} \in E, 1 \leq i \leq 3),
\end{align*}

We now provide an intuitive reading of this program.

- A collection of choice rules for each vertex $v$ captured by the first line in (11) intuitively says that vertex $v$ may be assigned some colors. 

(Condition 1)
Figure 2: Logic program for 3-coloring for graph $G_1$.

- The second line states that it is impossible that a vertex is not assigned a color. \textit{(Condition 1)}
- The third line says that it is impossible that a vertex is assigned two colors. \textit{(Condition 2)}
- The fourth line states that it is impossible that any two adjacent vertices are assigned the same color. \textit{(Condition 3)}

Recall graph $G_1$ in Figure 1. Figure 2 presents a logic program in spirit of (11) for $G_1$. Horizontal lines separate the clauses that come from distinct "schematic rules" in (11). This program has six answer sets including

\[
\{c_{a1}, c_{b2}, c_{c1}, c_{d3}\},
\]

which captures solution $S_{G_1}$. To encode the 3-coloring problem for graph $G_2$ one has to extend the set of rules in Figure 2 with rules

\[
\leftarrow c_{a1}, c_{c1} \quad \leftarrow c_{a2}, c_{c2} \quad \leftarrow c_{a3}, c_{c3}.
\]

This program has no answer sets, which captures the fact that this graph has no 3-colorings.

The ASP specification (11) illustrates the use of the so-called \textsc{generate} and \textsc{test} methodology within answer set programming paradigm. The
Figure 3: Common Architecture of ASP Systems

GENERATE part of the specification “defines” a collection of “perspective” answer sets that can be seen as potential solutions. The TEST part consists of conditions that eliminate the “perspective” answer sets of the GENERATE part that do not correspond to solutions. The first line in (11) corresponds to GENERATE: saying that any subset of the atoms of the form \( c_{vi} \ (v \in V \text{ in a given graph } (V,E) \text{ and } i \in \{1, 2, 3\}) \) forms a potential solution. The remaining lines correspond to TEST. Observe, how choice rules provide a convenient tool in ASP for formulating GENERATE, whereas constraints are used to formulate TEST.

3 Programs with Variables and Grounding

ASP practitioners develop applications that rely on languages, which go beyond propositional/ground atoms. Figure 3 presents a typical architecture of an answer set programming tool that encompasses two parts: a system called grounder and a system called solver. The former is responsible for eliminating variables in a program. The latter is responsible for finding answer sets of a respective propositional (ground) program. For instance, system GRINGO\(^2\) is a well known grounder that serve as front-ends for many solvers including CLASP. A combination of GRINGO and CLASP is known as system CLINGO.

We recall that given a signature \( \sigma \) consisting of object constants, variables, predicate symbols, and function symbols (where predicate and function symbols are associated with a nonnegative integer \( n \) called arity, so that we identify function symbols of arity 0 with object constants),

1. Any object constant or variable in \( \sigma \) is a \textit{term}, and
   
an expression of the form \( f(t_1, \ldots, t_n) \) is a \textit{term} where \( f \) is a function symbol in \( \sigma \) of arity \( n > 0 \) and \( t_1, \ldots, t_n \) are terms.

2. Any predicate symbol in \( \sigma \) of arity 0 is an atom, and

\(^2\)http://potassco.sourceforge.net/
an expression of the form $p(t_1,\ldots,t_n)$ is an atom where $p$ is a predicate symbol in $\sigma$ of arity $n$ and $t_1,\ldots,t_n$ are terms.

3. Any term that contains no variables is called ground. Similarly, any atom that contains no variables is ground. Otherwise, we refer to these entities as non-ground.

A logic program with variables is a finite set of rules of the form (1), where $a_0$ is symbol $\bot$ or a non-ground atom; and $a_1,\ldots,a_p$ are non-ground atoms. Grounding a logic program replaces each rule with all its instances obtained by substituting ground terms, formed from the object constants and function symbols occurring in the program, for all variables. For a program $\Pi$, by $\text{ground}(\Pi)$ we denote the result of its grounding. (We use the convention common in logic programming: variables are represented by capitalized identifiers.) We illustrate this concept on an example. Let $\Pi$ be a program

$$
\begin{align*}
\{a(1)\} & \{a(2)\} \{b(1)\} \\
c(X) & \leftarrow a(X), b(X),
\end{align*}
$$

$\text{ground}(\Pi)$ follows

$$
\begin{align*}
\{a(1)\} & \{a(2)\} \{b(1)\} \\
c(1) & \leftarrow a(1), b(1) \\
c(2) & \leftarrow a(2), b(2).
\end{align*}
$$

The answer sets of a program $\Pi$ with variables are answer sets of $\text{ground}(\Pi)$. For instance, there are eight answer sets of program (12) including $\emptyset$ and set $\{a(1) \, b(1) \, c(1)\}$.

**Problem 1.** (a) Consider the following program with variables

$$
\begin{align*}
\{a(1)\} & \{a(2)\} \{b(1)\} \\
ad(X,Y) & \leftarrow a(X), b(Y).
\end{align*}
$$

Construct the result of grounding for this program.

(b) Follow the link [https://potassco.org/clingo/run/](https://potassco.org/clingo/run/). Replace symbol “$\leftarrow$“ by ”:-“ in the ground program you constructed in (a) and let CLINGO run on your program using reasoning mode ”enumerate all“.

(c) Using the same procedure as in (b), find all answer sets for the program listed in (a). Do answer sets found in (b) coincide with the ones enumerated in this step?

Given a program $\Pi$ with variables, grounders often produce a variable-free program that is smaller than $\text{ground}(\Pi)$, but still has the same answer
sets as ground(Π); we call any such program an image of Π. For example, program

\[
\begin{align*}
\{a(1)\} & \{a(2)\} \{b(1)\} \\
c(1) & \leftarrow a(1), b(1)
\end{align*}
\]

is an image of (12).

Passing parameter –t in command line when calling CLINGO on a program will force the system to produce ground program in human readable form. Running CLINGO in online interface with checkbox ”statistics” allows one to obtain valuable information about the execution of the system. For example, lines titled

- “Rules” provides number of rules in a grounding of the input and suggests the relative size of a ground program,
- “Choices” corresponds to the number of backtracks done by the system in search for the solution (we will learn of this feature in a little bit),
- “Time” reports the execution time of the system.

In command line to obtain statistics while running CLINGO use flag “--stats”

Problem 2. Let CLINGO run on the program (13) using reasoning mode ”enumerate all“ and marking checkbox ”statistics“. What is the number of rules that CLINGO reports?

This number corresponds to the size of the image (measured in number of rules) produced by GRINGO after grounding the input program.

Think of an image for program (13) that is of the same size as GRINGO computed. List this image.

When a program Π with variables has at least one function symbol and at least one object constant, grounding results in infinite ground(Π). Yet, even for an input program of this kind, grounders often find an image that is a finite propositional program (finite image). For instance, for program

\[
\begin{align*}
p(0) \\
q(f(X)) & \leftarrow p(X)
\end{align*}
\]

(14)
grounding results in infinite program outlined below

\[
\begin{align*}
p(0) \\
q(f(0)) & \leftarrow p(0) \\
q(f(f(0))) & \leftarrow p(f(0)) \\
q(f(f(f(0)))) & \leftarrow p(f(f(0))) \\
& \cdots
\end{align*}
\]
A finite image of program (14) follows

\[ p(0) \quad q(f(0)) \leftarrow p(0). \]

Program

\[
\begin{align*}
    & p(0) \\
    & q(f(0))
\end{align*}
\]

is another image of (14). In fact, given program (14) as an input grounder GRINGO will generate the latter image.

To produce images for input programs, grounders follow techniques exemplified by intelligent grounding. Different grounders implement distinct procedures so that they may generate different images for the same input program. One can intuitively measure the quality of a produced image by its size so that the smaller the image is the better. A common syntactic restriction that grounders pose on input programs is “safety”. A program \( \Pi \) is safe if every variable occurring in a rule of \( \Pi \) also occurs in positive body of that rule. For instance, programs (12) and (14) are safe. The safety requirement suggests that positive body of a rule must contain information on the values that should be substituted for a variable in the process of grounding. Safety is instrumental in designing grounding techniques that utilize knowledge about the structure of a program for constructing smaller images. The GRINGO grounder and the grounder of the DLV system expect programs to be safe. For programs with function symbols, to guarantee that the grounding process terminates, grounders pose additional syntactic restrictions (in other words, to guarantee that a grounder is able to construct a finite image).

4 Formalization of Graph coloring problem by means of a logic program with variables

We now revisit our graph coloring example and illustrate how often a set of propositional rules that follow a simple pattern can be represented concisely by means of logic programs with variables. Recall program (11). We now capture atoms of the form \( c_{vi} \) by expressions \( c(v, i) \), where \( c \) is a predicate symbol and \( v, i \) are object constants denoting a vertex \( v \) and color \( i \) respectively. Atom of the form \( vtx(v) \), intuitively, states that an object constant \( v \) is a vertex, while atom \( e(v, w) \) states that there is an edge from vertex \( v \) to vertex \( w \) in a given graph. Atom \( color(i) \) states that an object
constant \( i \) represents a color. Recall graph coloring problem \( GC \) for an input graph \((V, E)\). We now present a program with variables that encodes a solution to this problem. First, this program consists of facts that encode graph \((V, E)\):

\[
\begin{align*}
vtx(v) & \quad (v \in V) \\
e(v, w) & \quad \{\{v, w\} \in E}
\end{align*}
\]

(15)

Second, facts

\[
\text{color}(c) \quad (c \in \{1, 2, 3\})
\]

(16)

enumerate three colors of the problem. The following rules conclude the description of the program:

\[
\begin{align*}
\{c(V, I)\} & \leftarrow vtx(V), \text{color}(I) \\
\leftarrow & \text{not } c(V, 1), \text{not } c(V, 2), \text{not } c(V, 3), vtx(V) \\
\leftarrow & c(V, I), c(V, J), I < J, vtx(V), \text{color}(I), \text{color}(J) \\
\leftarrow & c(V, I), c(W, I), vtx(V), vtx(W), \text{color}(I), e(V, W)
\end{align*}
\]

(17) (18) (19) (20)

These rules are the counterparts of groups of rules in propositional program (11). Indeed,

- rule (17) states that every vertex may be assigned some colors;
- the second rule (18) states that it is impossible that a vertex is not assigned a color;
- rule (19) says that it is impossible that a vertex is assigned two colors; and
- rule (20) says that it is impossible that any two adjacent vertexes are assigned the same color.

Programs with variables permit for a concise encoding of an instance of a search problem. Indeed, size of a program composed of rules (15-18) is almost identical to the size of a given graph \((V, E)\). There are \(|V| + |E| + 7\) rules in this program. On the other hand, the line

\[
\leftarrow c_{vi}, c_{wi} \quad (\{v, w\} \in E, 1 \leq i \leq 3)
\]

of program (11) alone encapsulates \(3|E|\) rules.
5 Modeling of Search Problems in ASP

Answer set programming provides a general purpose modeling language that supports elaboration tolerant solutions for search problems. We now define a search problem abstractly. A search problem \( P \) consists of a set of instances with each instance \( I \) assigned a finite set \( S_P(I) \) of solutions. In answer set programming to solve a search problem \( P \), we construct a program \( \Pi_P \) that captures problem specifications so that when extended with facts \( D_I \) representing an instance \( I \) of the problem, the answer sets of \( \Pi_P \cup D_I \) are in one to one correspondence with members in \( S_P(I) \). In other words, answer sets describe all solutions of problem \( P \) for the instance \( I \). Thus solving of a search problem is reduced to finding a uniform encoding of its specifications by means of a logic program with variables.

For example, an instance of the graph coloring search problem \( GC \) is a graph. All 3-colorings for a given graph form its solutions set. Consider any graph \((V,E)\). By \( D_{(V,E)} \) we denote facts in (15) that encode graph \((V,E)\). By \( \Pi_{gc} \) we denote a program composed of rules in (16-18). This program captures specifications of 3-coloring problem so that answer sets of \( \Pi_{gc} \cup D_{(V,E)} \) correspond to solutions to instance graph \((V,E)\) of a problem.

Recall graphs \( G_1 \) and \( G_2 \) presented in Figure 1. Facts \( D_{G_1} \) representing \( G_1 \) follow

\[
\text{vtx}(a) \hspace{1em} \text{vtx}(b) \hspace{1em} \text{vtx}(c) \hspace{1em} \text{vtx}(d) \hspace{1em} e(a,b) \hspace{1em} e(b,c) \hspace{1em} e(c,d) \hspace{1em} e(d,a) \hspace{1em} e(b,d).
\]

Program \( \Pi_{gc} \cup D_{G_1} \) has six answer sets, including

\[
\{\text{vtx}(a) \hspace{1em} \text{vtx}(b) \hspace{1em} \text{vtx}(c) \hspace{1em} \text{vtx}(d) \\
\text{e}(a,b) \hspace{1em} \text{e}(b,c) \hspace{1em} \text{e}(c,d) \hspace{1em} \text{e}(d,a) \hspace{1em} \text{e}(b,d) \\
\text{color}(1) \hspace{1em} \text{color}(2) \hspace{1em} \text{color}(3) \\
\text{c}(a,1) \hspace{1em} \text{c}(b,2) \hspace{1em} \text{c}(c,1) \hspace{1em} \text{c}(d,3)\},
\]

which captures solution \( S_{G_1} \). Similarly, we can use encoding \( \Pi_{gc} \) to establish whether 3-colorings exist for graph \( G_2 \). Facts \( D_{G_2} \) representing \( G_2 \) consists of facts in \( D_{G_1} \) and an additional fact \( e(a,c) \). Program \( \Pi_{gc} \cup D_{G_2} \) has no answer sets suggesting that no 3-colorings exist for graph \( G_2 \).

It is important to mention that the languages supported by ASP grounders and solvers go beyond rules presented here. For instance, GRINGO versions 4.5+ support such constructs as aggregates, cardinality expressions, intervals, pools. It is beyond the scope of this lecture to formally discuss these constructs, but it is worth mentioning that they generally allow us more concise, intuitive, and elaboration tolerant encodings of problems.
Also, they often permit to utilize more sophisticated and efficient procedures in solving.

For instance, a single rule

\[ \leftarrow \text{not } 1\#\text{count}\{V, I : c(V, I)\}1, \text{vtx}(V). \]  

(21)
can replace two rules (18) and (19) in program \( \Pi_{ge} \). This rule states that

it must be the case that a vertex is assigned exactly one color.  

(22)
This shorter program will result in smaller groundings for instances of the \( GC \) problem paving the way to more efficient solving. Cardinality construct

\[ 1\#\text{count}\{V, I : c(V, I)\}1 \]

intuitively suggests us to count, for a given value of \( V \) the tuples \( (V, I) \) for which atom of the form \( c(V, I) \) belongs to the answer set. Number 1 to the right and to the left of this aggregate expression tells us a specific condition on the count, in particular, that it has to be exactly 1. Indeed, the number to the right suggests at most count whereas the number to the left suggest at least count. Cardinality expressions form only one example of multitude of constructs that GRINGO language offers for effective modeling of problem specifications.

It is worth noting that the GRINGO language also provides syntactic sugar, i.e., convenient abbreviations for groups of rules. For example, expression

\[ \text{edge}(a, b; b; c; c, d). \]

abbreviates the set of facts

\[ \text{edge}(a, b). \text{edge}(b, c). \text{edge}(c, d). \]

Expression

\[ 1\{c(V, I) : \text{color}(I)\} \leftarrow \text{vtx}(V). \]

abbreviates rules (17) and (18). So that we can encode Condition 1 listed in Section 2 by one rule.

Expression

\[ 1\{c(V, I) : \text{color}(I)\}1 \leftarrow \text{vtx}(V). \]  

(23)
abbreviates a collection of two rules (17) and (21). Intuitively, we can read the meaning of this rule as stated in (22) that we used to characterize an intuitive meaning of constraint (21). Although English statements turn out
to be the same for rules (21) and (23), the formal meaning of mathematical expression (23) obviously extends that of (21). In addition to a constraint on solutions captured by (21) [which is one of the rules abbreviated by expression (23)], choice rule (17) [the other rule abbreviated by expression (23)] provides ground for atoms of the form $c(\cdot, \cdot)$ be part of solutions. Or, in other words, choice rule (17) removes closed world assumption from $c(\cdot, \cdot)$ atoms.

**Problem 3.** Note that directive #show c/2. added to a CLINGO program $\Pi_{gc}$ allows one to instruct CLINGO to only output these atoms in the computed answer sets that have the form $c(\cdot, \cdot)$.

(a) Recall the statement of graph coloring problem. Imagine the following extension to that statement: there is at most one node in a given graph colored by color named 1. Extend the program $\Pi_{gc}$ so that this new statement is respected.

(b) For graph $G_1$, how many solutions to the new graph coloring problem stated in (a) are there?

(c) Use CLINGO to test your solution in by running it on the program $\Pi_{gc} \cup D_{G_1}$ extended with the code you developed in (a). List the answer sets that CLINGO computes (list only the atoms of the form $c(\cdot, \cdot)$).

## 6 The Generate-Define-and-Test Modeling Methodology of ASP

Previously, we presented how the GENERATE and TEST methodology is applicable within answer set programming. Yet, an essential feature of logic programs is their ability to elegantly and concisely “define” predicates. Logic programs provide a convenient language for expressing inductive/recursive definitions.

The GENERATE, DEFINE, and TEST is a typical methodology used by ASP practitioners in designing programs. It generalizes the GENERATE and TEST methodology discussed earlier. The roles of the GENERATE and TEST parts of a program stay the same so that, informally,

- **GENERATE** defines a large collection of answer sets that could be seen as potential solutions, while

- **TEST** consists of rules that eliminate the answer sets of the GENERATE part that do not correspond to solutions.

- The DEFINE section expresses additional, auxiliary concepts and connects the GENERATE and TEST parts.
To illustrate the essence of define, consider a Hamiltonian cycle search problem:

Given a directed graph \((V, E)\), the goal is to find a Hamiltonian cycle — a set of edges that induce in \((V, E)\) a directed cycle going through each vertex exactly once.

This is an important combinatorial search problem related to Traveling Salesperson problem. A solution can be described by a set of atoms of the form \(in(v, w), (v, w) \in E\) of the given graph \((V, E)\); including \(in(v, w)\) in the set indicates that an edge from vertex \(v\) to vertex \(w\) is part of a found Hamiltonian cycle. A solution is a set \(X\) satisfying the following conditions

1. the Hamiltonian cycle is formed by the edges of a given graph \((V, E)\), or, in other words, the extension of predicate \(in\) is a subset of \(E\),

2. \(X\) does not contain a pair of different atoms of the form \(in(u, v), in(u, v')\) (two selected edges start at the same vertex; thus we pass node \(u\) only once on the way out),

3. \(X\) does not contain a pair of different atoms of the form \(in(u, v), in(u', v)\) (two selected edges end at the same vertex; thus we pass node \(u\) only once on the way in),

4. For each pair \(u, v\) of vertexes, \(X\) is such that \((u, v)\) is a part of the transitive closure of \(in\) relation defined by \(X\). (a found subset of edges of the graph is indeed a cycle.) Recall that

   - the transitive closure of a binary relation \(R\) (in our case relation \(in\)) on a set \(Y\) of elements (in our case the set of the vertexes of the given graph) is the smallest relation on \(Y\) that contains \(R\) and is transitive.

   - a transitive relation \(R\) on set \(Y\) of elements is such that for any three elements \(a, b, c\) (not necessarily distinct elements) in \(Y\) the following property holds if \(a\) is in relation \(R\) with \(b\) and \(b\) is in relation \(R\) with \(c\) then \(a\) is in relation \(R\) with \(c\).

We now formalize the specifications of Hamiltonian cycle problem by means of generate, define, and test methodology. As in the encoding \(\Pi_{gc}\) of the graph coloring problem, we use expressions of the form \(vtx(v)\) and \(e(v, w)\) to encode an input graph.
Choice rule

\[
\{ \text{in}(X,Y) \} \leftarrow \text{e}(X,Y)
\]  (24)

forms the GENERATE part of the problem. This rule states that any subset of edges of a given graph may form a Hamiltonian cycle (\textit{Condition 1}). Answer sets of a program composed of this rule and a set of facts encoding an input graph will correspond to all subsets of edges of the graph. For instance, program composed of facts \(D_{G_1}\) that encode directed graph \(G_1\) introduced in Figure 1 extended by rule (24) has 32 answer sets each representing a different subset of its edges.

The remaining \textit{Conditions 2-4} are captured in the TEST part. To formulate \textit{Condition 4}, an auxiliary concept of \textit{reachable} (a transitive closure of \textit{in}) is required so that we can capture the restriction that a found subset of edges of the graph is also a cycle. The \texttt{define} part follows

\[
\begin{align*}
\text{reachable}(V,V) & \leftarrow \text{vtx}(V) \\
\text{reachable}(U,W) & \leftarrow \text{in}(U,V), \text{reachable}(V,W), \text{vtx}(U), \text{vtx}(V), \text{vtx}(W)
\end{align*}
\]  (25)

These rules define \textit{transitive closure} of the predicate \textit{in}: all pairs of vertexes \((u,v)\) such that \(v\) can be reached from \(u\) by following zero or more edges that are \textit{in}.

We are now ready to state the TEST part composed of three rules

\[
\begin{align*}
\text{Condition 1} & \leftarrow \text{in}(U,V), \text{in}(U,W), V \neq W, \text{vtx}(U;V;W) \\
\text{Condition 2} & \leftarrow \text{in}(U,V), \text{in}(W,V), U \neq W, \text{vtx}(U;V;W) \\
\text{Condition 3} & \leftarrow \text{not reachable}(U,V), \text{vtx}(U;V).
\end{align*}
\]  (26)

Rules (24), (25), and (26) form a program \(\Pi_{hc}\) that captures specifications of Hamiltonian cycle search problem. Extending \(\Pi_{hc}\) with facts representing a directed graph results in a program whose answer sets describe all Hamiltonian cycles of this graph. For example, program \(\Pi_{hc} \cup D_{G_1}\) has only one answer set. Set

\[
\{ \text{in}(a,b) \text{ in}(b,c) \text{ in}(c,d) \text{ in}(d,a) \}
\]

contains all \textit{in}-edges of that answer set stating that edges \((a,b), (b,c), (c,d), \) and \((d,a)\) form the only Hamiltonian cycle for graph \(G_1\).

Concise encoding of transitive closure is a feature of answer set programming that constitutes an essential difference between ASP and formalisms based on classical logic. For example, transitive closure is not expressible by first-order formulas. Thus any subset of first-order logic taken as the
language with variables for modeling search problems declaratively will fail at defining (directly) concepts that rely on transitive closure.

In mastering the art of answer set programming it is enough to develop intuitions about answer sets of the programs with variables that are formed in accordance with the generate, define, and test methodology. Some of these intuitions will stem from the general properties you encountered earlier such as any element of answer set must appear in the head of some rule in a program. For the general definition of an answer set, it is more difficult to develop intuitions on what answer sets conceptually are and unnecessary.

**Problem 4.** Recall that the rule

\[ \leftarrow \text{in}(U,V), \text{in}(U,W), V \neq W, \text{vtx}(U;V;W) \]

in \( \Pi_{hc} \) states that no two selected edges start at the same node. Rewrite this rule using aggregate \#count exemplified in rule (21). You may wish to consult Clingo-Gringo manual Section 3.1.12 for more details on aggregates.

### 7 ASP Formulation of \( n \)-Queens

We now turn our attention to another combinatorial search problem: \( n \)-queens problem.

The goal is to place \( n \) queens on an \( n \times n \) chessboard so that no two queens would be placed on the same row, column, and diagonal.

A solution can be described by a set of atoms of the form \( q(i,j) \) (1 \( \leq i, j \leq n \)); including \( q(i,j) \) in the set indicates that there is a queen at position \( (i,j) \). A solution is a set \( X \) satisfying the following conditions:

1. the cardinality of \( X \) is \( n \),

2. \( X \) does not contain a pair of different atoms of the form \( q(i,j), q(i',j) \) (two queens on the same row),

3. \( X \) does not contain a pair of different atoms of the form \( q(i,j), q(i,j') \) (two queens on the same column),

4. \( X \) does not contain a pair of different atoms of the form \( q(i,j), q(i',j') \) with \( |i' - i| = |j' - j| \) (two queens on the same diagonal).

Here is the representation of this program in the input language of CLINGO:
number(1..n).

%CCondition 1 and 2
1{q(K,J) : number(K)\}1:- number(J).

%CCondition 3
:- q(I,J), q(I,J1), J<J1.

%CCondition 4
:- q(I,J), q(I1,J1), J<J1, |I1-I|=J1-J.

We name this program queens.clingo.
  Appending the line
   # const n=8.
   to the code in queens.clingo will instruct answer set system CLINGO to search
for solution for 8-queens problem. Alternatively, the command line
   clingo -c n=8 queens.clingo
   instructs the answer set system CLINGO to find a single solution for 8-queens
problem, whereas the command line
   clingo -c n=8 queens.clingo 0
   instructs CLINGO to find all solutions to 8-queens program. The command line
   gringo -c n=8 queens.clingo > queens.8.grounded
   instructs the grounder GRINGO to ground 8-queens problem; the ground
problem (ready for processing with CLASP) is stored in file queens.8.grounded.
The command lines
   gringo -t -c n=8 queens.clingo
   or
   clingo -t -c n=8 queens.clingo
   will produce human-readable grounded 8-queens problem.
   The command line
   clasp < queens.8.grounded
will instruct the answer set solver CLASP to look for answer sets of a program in *queens.8.grounded*.

An extract from the output of the last command line follows

```
...  
Answer: 92 
number(1) number(2) number(3) number(4)  
number(5) number(6) number(7) number(8)  
q(5,8) q(7,7) q(2,6) q(6,5) q(3,4) q(1,3) q(8,2) q(4,1)  
SATISFIABLE  
```

This 92nd solution found by the solver encodes the following valid configuration of queens on the board

```
1  2  3  4  5  6  7  8  
1 Q  
2 Q  
3 Q  
4Q  
5 Q  
6 Q  
7 Q  
8 Q  
```

Similarly, appending the line

```
# const n=4.  
```

to the code in *queens.clingo* will instruct CLINGO to solve 4-queens problem. The command line

```
clingo -c n=4 queens.clingo 0  
```

instructs CLINGO to find all solutions for 4-queens problem.

**Problem 5.** (a) Use CLINGO to find all solutions to the 8-queens problem that have a queen at (1,1). How many solutions of the kind are there?

(b) Use CLINGO to find all solutions to the 12-queens problem that have a queen at (1,1). How many solutions of the kind are there?

Submit the lines of code that you wrote to solve these problems.

**Problem 6.** (a) Use CLINGO to find all solutions to the 8-queens problem that have no queens in the 4 × 4 square in the middle of the board. How many solutions of the kind are there?
(b) Use clingo to find all solutions to the 10-queens problem that have no queens in the 4×4 square in the middle of the board. How many solutions of the kind are there?

Submit the lines of code that you wrote to solve these problems.

Acknowledgments


References